

NAG Library Routine Document

C09FCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

C09FCF computes the three-dimensional multi-level discrete wavelet transform (DWT). The initialization routine C09ACF must be called first to set up the DWT options.

2 Specification

```
SUBROUTINE C09FCF (M, N, FR, A, LDA, SDA, LENC, C, NWL, DWTLVM, DWTLVN, &
                  DWTLVFR, ICOMM, IFAIL)
INTEGER          M, N, FR, LDA, SDA, LENC, NWL, DWTLVM(NWL), &
                  DWTLVN(NWL), DWTLVFR(NWL), ICOMM(260), IFAIL
REAL (KIND=nag_wp) A(LDA,SDA,FR), C(LENC)
```

3 Description

C09FCF computes the multi-level DWT of three-dimensional data. For a given wavelet and end extension method, C09FCF will compute a multi-level transform of a three-dimensional array A , using a specified number, n_{fwd} , of levels. The number of levels specified, n_{fwd} , must be no more than the value l_{max} returned in NWLMAX by the initialization routine C09ACF for the given problem. The transform is returned as a set of coefficients for the different levels (packed into a single array) and a representation of the multi-level structure.

The notation used here assigns level 0 to the input data, A . Level 1 consists of the first set of coefficients computed: the seven sets of detail coefficients are stored at this level while the approximation coefficients are used as the input to a repeat of the wavelet transform at the next level. This process is continued until, at level n_{fwd} , all eight types of coefficients are stored. All coefficients are packed into a single array.

4 References

Wang Y, Che X and Ma S (2012) Nonlinear filtering based on 3D wavelet transform for MRI denoising *URASIP Journal on Advances in Signal Processing* **2012:40**

5 Parameters

- 1: M – INTEGER *Input*
On entry: the number of rows of each two-dimensional frame.
Constraint: this must be the same as the value M passed to the initialization routine C09ACF.
- 2: N – INTEGER *Input*
On entry: the number of columns of each two-dimensional frame.
Constraint: this must be the same as the value N passed to the initialization routine C09ACF.
- 3: FR – INTEGER *Input*
On entry: the number of two-dimensional frames.
Constraint: this must be the same as the value FR passed to the initialization routine C09ACF.

- 4: A(LDA, SDA, FR) – REAL (KIND=nag_wp) array *Input*
On entry: the m by n by fr three-dimensional input data A , where with A_{ijk} stored in $A(i, j, k)$.
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which C09FCF is called.
Constraint: $LDA \geq M$.
- 6: SDA – INTEGER *Input*
On entry: the second dimension of the array A as declared in the (sub)program from which C09FCF is called.
Constraint: $SDA \geq N$.
- 7: LENC – INTEGER *Input*
On entry: the dimension of the array C as declared in the (sub)program from which C09FCF is called.
Constraint: $LENC \geq n_{ct}$, where n_{ct} is the total number of wavelet coefficients that correspond to a transform with NWL levels.
- 8: C(LENC) – REAL (KIND=nag_wp) array *Output*
On exit: the coefficients of the discrete wavelet transform. If you need to access or modify the approximation coefficients or any specific set of detail coefficients then the use of C09FYF or C09FZF is recommended. For completeness the following description provides details of precisely how the coefficients are stored in C but this information should only be required in rare cases.
Let $q(i)$ denote the number of coefficients of each type at level i , for $i = 1, 2, \dots, n_{fwd}$, such that $q(i) = DWTLVM(n_{fwd} - i + 1) \times DWTLVN(n_{fwd} - i + 1) \times DWTLVFR(n_{fwd} - i + 1)$. Then, letting $k_1 = q(n_{fwd})$ and $k_{j+1} = k_j + q(n_{fwd} - \lceil j/7 \rceil + 1)$, for $j = 1, 2, \dots, 7n_{fwd}$, the coefficients are stored in C as follows:
C(i), for $i = 1, 2, \dots, k_1$
Contains the level n_{fwd} approximation coefficients, $a_{n_{fwd}}$. Note that for computational efficiency reasons these coefficients are stored as $DWTLVM(1) \times DWTLVN(1) \times DWTLVFR(1)$ in C .
C(i), for $i = k_j + 1, \dots, k_{j+1}$
Contains the level $n_{fwd} - \lceil j/7 \rceil + 1$ detail coefficients. These are:
LLH coefficients if $j \bmod 7 = 1$;
LHL coefficients if $j \bmod 7 = 2$;
LHH coefficients if $j \bmod 7 = 3$;
HLL coefficients if $j \bmod 7 = 4$;
HLH coefficients if $j \bmod 7 = 5$;
HHL coefficients if $j \bmod 7 = 6$;
HHH coefficients if $j \bmod 7 = 0$,
for $j = 1, \dots, 7n_{fwd}$. See Section 2.1 in the C09 Chapter Introduction for a description of how these coefficients are produced.
Note that for computational efficiency reasons these coefficients are stored as $DWTLVFR(\lceil j/7 \rceil) \times DWTLVM(\lceil j/7 \rceil) \times DWTLVN(\lceil j/7 \rceil)$ in C .

- 9: NWL – INTEGER *Input*
On entry: the number of levels, n_{fwd} , in the multi-level resolution to be performed.
Constraint: $1 \leq \text{NWL} \leq l_{\text{max}}$, where l_{max} is the value returned in NWLMAX (the maximum number of levels) by the call to the initialization routine C09ACF.
- 10: DWTLVM(NWL) – INTEGER array *Output*
On exit: the number of coefficients in the first dimension for each coefficient type at each level. DWTLVM(i) contains the number of coefficients in the first dimension (for each coefficient type computed) at the $(n_{\text{fwd}} - i + 1)$ th level of resolution, for $i = 1, 2, \dots, n_{\text{fwd}}$.
- 11: DWTLVN(NWL) – INTEGER array *Output*
On exit: the number of coefficients in the second dimension for each coefficient type at each level. DWTLVN(i) contains the number of coefficients in the second dimension (for each coefficient type computed) at the $(n_{\text{fwd}} - i + 1)$ th level of resolution, for $i = 1, 2, \dots, n_{\text{fwd}}$.
- 12: DWTLVFR(NWL) – INTEGER array *Output*
On exit: the number of coefficients in the third dimension for each coefficient type at each level. DWTLVFR(i) contains the number of coefficients in the third dimension (for each coefficient type computed) at the $(n_{\text{fwd}} - i + 1)$ th level of resolution, for $i = 1, 2, \dots, n_{\text{fwd}}$.
- 13: ICOMM(260) – INTEGER array *Communication Array*
On entry: contains details of the discrete wavelet transform and the problem dimension as setup in the call to the initialization routine C09ACF.
On exit: contains additional information on the computed transform.
- 14: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, FR = $\langle \text{value} \rangle$.

Constraint: FR = $\langle \text{value} \rangle$, the value of FR on initialization (see C09ACF).

On entry, M = $\langle \text{value} \rangle$.

Constraint: M = $\langle \text{value} \rangle$, the value of M on initialization (see C09ACF).

On entry, N = $\langle \text{value} \rangle$.

Constraint: N = $\langle \text{value} \rangle$, the value of N on initialization (see C09ACF).

IFAIL = 2

On entry, LDA = $\langle value \rangle$ and M = $\langle value \rangle$.
Constraint: LDA \geq M.

On entry, SDA = $\langle value \rangle$ and N = $\langle value \rangle$.
Constraint: SDA \geq N.

IFAIL = 3

On entry, LENC = $\langle value \rangle$.
Constraint: LENC \geq $\langle value \rangle$, the total number of coefficients to be generated.

IFAIL = 5

On entry, NWL = $\langle value \rangle$.
Constraint: NWL \geq 1.

On entry, NWL = $\langle value \rangle$ and NWLMAX = $\langle value \rangle$ in C09ACF.
Constraint: NWL \leq NWLMAX in C09ACF.

IFAIL = 6

Either the communication array ICOMM has been corrupted or there has not been a prior call to the initialization routine C09ACF.

The initialization routine was called with WTRANS = 'S'.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.
See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The accuracy of the wavelet transform depends only on the floating-point operations used in the convolution and downsampling and should thus be close to *machine precision*.

8 Parallelism and Performance

Not applicable.

9 Further Comments

The example program shows how the wavelet coefficients at each level can be extracted from the output array C. Denoising can be carried out by applying a thresholding operation to the detail coefficients at every level. If c_{ij} is a detail coefficient then $\hat{c}_{ij} = c_{ij} + \sigma\epsilon_{ij}$ and $\sigma\epsilon_{ij}$ is the transformed noise term. If some threshold parameter α is chosen, a simple hard thresholding rule can be applied as

$$\bar{c}_{ij} = \begin{cases} 0, & \text{if } |\hat{c}_{ij}| \leq \alpha \\ \hat{c}_{ij}, & \text{if } |\hat{c}_{ij}| > \alpha, \end{cases}$$

taking \bar{c}_{ij} to be an approximation to the required detail coefficient without noise, c_{ij} . The resulting coefficients can then be used as input to C09FDF in order to reconstruct the denoised signal. See Section 10 in C09FZF for a simple example of denoising.

See the references given in the introduction to this chapter for a more complete account of wavelet denoising and other applications.

10 Example

This example computes the three-dimensional multi-level discrete wavelet decomposition for $7 \times 6 \times 5$ input data using the biorthogonal wavelet of order 1.1 (set WAVNAM = 'BIOR1.1' in C09ACF) with periodic end extension, prints a selected set of wavelet coefficients and then reconstructs and verifies that the reconstruction matches the original data.

10.1 Program Text

```

Program c09fcfe

!      C09FCF Example Program Text
!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: c09acf, c09fcf, c09fdf, c09fyf, nag_wp, x02ajf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: eps, esq, frob
Integer                    :: fr, i, ifail, j, k, lda, ldb, ldd,   &
                             lenc, m, n, nf, nwcfr, nwcm, nwcnc,   &
                             nwct, nwl, nwlinv, nwlmax, sda, sdb, &
                             sdd, want_coeffs, want_level
Character (10)              :: mode, wavnam, wtrans
Character (33)              :: title
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,:,:), b(:,:,:), c(:), d(:,:,:), &
                             e(:,:,:)
Integer, Allocatable        :: dwtlvfr(:), dwtlvm(:), dwtlvn(:)
Integer                    :: icomm(260)
Character (3)               :: cpass(0:7)
!      .. Intrinsic Procedures ..
Intrinsic                   :: max, real, sqrt
!      .. Executable Statements ..
Continue
Write (nout,*) 'C09FCF Example Program Results'
Write (nout,*)
!      Skip heading in data file
Read (nin,*)
!      Read problem parameters
Read (nin,*) m, n, fr
Read (nin,*) wavnam, mode
lda = m
sda = n
ldb = m
sdb = n
Allocate (a(lda,sda,fr),b(ldb,sdb,fr),e(m,n,fr))

Write (nout,99999) wavnam, mode, m, n, fr

!      Read data array and write it out

Do j = 1, fr
  Do i = 1, m
    Read (nin,*) a(i,1:n,j)
  End Do
  If (j<fr) Read (nin,*)
End Do

```

```

Write (nout,*) ' Input Data      A : '
Do j = 1, fr
  Write (nout,99997) j
  Do i = 1, m
    Write (nout,99998) a(i,1:n,j)
  End Do
End Do

! Query wavelet filter dimensions
! For Multi-Resolution Analysis, decomposition, wtrans = 'M'
wtrans = 'Multilevel'
ifail = 0
Call c09acf(wavnam,wtrans,mode,m,n,fr,nwlmax,nf,nwct,nwcn,nwcf,icomm, &
  ifail)

lenc = nwct
Allocate (c(lenc),dwtlvm(nwlmax),dwtlvn(nwlmax),dwtlvfr(nwlmax))

nwl = nwlmax

! Perform Discrete Wavelet transform
ifail = 0
Call c09fcf(m,n,fr,a,lda,sda,lenc,c,nwl,dwtlvm,dwtlvn,dwtlvfr,icomm, &
  ifail)

Write (nout,99996) nwl
Write (nout,99995)
Write (nout,99992) dwtlvm(1:nwl)
Write (nout,99994)
Write (nout,99992) dwtlvn(1:nwl)
Write (nout,99993)
Write (nout,99992) dwtlvfr(1:nwl)

! Print the first level HLL coefficients
want_level = 1
want_coefs = 4

nwcm = dwtlvm(nwl-want_level+1)
nwcn = dwtlvn(nwl-want_level+1)
nwcf = dwtlvfr(nwl-want_level+1)

! Allocate space to store the selected coefficients
ldd = nwcm
sdd = nwcn
Allocate (d(ldd,sdd,nwcf))

Write (nout,99987) want_level, nwcm, nwcn, nwcf

cpass(0:7) = ('LLL','LLH','LHL','LHH','HLL','HLH','HHL','HHH')
If (want_coefs==0) Then
  title = 'Approximation coefficients (LLL)'
Else
  title = 'Detail coefficients (' // cpass(want_coefs) // ')'
End If

! Extract coefficients
Call c09fyf(want_level,want_coefs,lenc,c,d,ldd,sdd,icomm,ifail)

! Print out the selected set of coefficients
Write (nout,99986) title
Write (nout,99989) want_level, want_coefs
Do k = 1, nwcf
  Write (nout,99988) k
  Do i = 1, nwcm
    Write (nout,99998) d(i,1:nwcn,k)
  End Do
End Do

nwlinv = nwl

```

```

!   Reconstruct original data
    ifail = 0
    Call c09fdf(nwlinv,lenc,c,m,n,fr,b,ldb,sdb,icomm,ifail)

!   Check reconstruction matches original
    eps = 10.0_nag_wp*real(m,kind=nag_wp)*real(n,kind=nag_wp)* &
        real(fr,kind=nag_wp)*x02ajf()

    e(1:m,1:n,1:fr) = b(1:m,1:n,1:fr) - a(1:m,1:n,1:fr)
    frob = 0.0_nag_wp
    Do k = 1, fr
        esq = 0.0_nag_wp
        Do j = 1, n
            Do i = 1, m
                esq = esq + e(i,j,k)**2
            End Do
        End Do
        frob = max(frob,sqrt(esq))
    End Do

    If (frob>eps) Then
        Write (nout,99991)
    Else
        Write (nout,99990)
    End If

99999 Format (1X,' MLDWT :: Wavelet : ',A/1X,'           End mode : ',A/1X, &
        '           M           : ',I4/1X,'           N           : ',I4/1X, &
        '           FR           : ',I4/)
99998 Format (8(F8.4,1X):)
99997 Format (1X,' Frame ',I2,' : ')
99996 Format (/1X,' Number of Levels : ',I10)
99995 Format (1X,' Number of coefficients in 1st dimension for each level :')
99994 Format (1X,' Number of coefficients in 2nd dimension for each level :')
99993 Format (1X,' Number of coefficients in 3rd dimension for each level :')
99992 Format (8(I8,1X):)
99991 Format (/1X,' Fail: Frobenius norm of B-A, where A is the original '/1X, &
        ' data and B is the reconstruction, is too large.')
99990 Format (/1X,' Success: the reconstruction matches the original.')
99989 Format (1X,' Level ',I2,', Coefficients ',I2,' : ')
99988 Format (1X,' Frame ',I2,' : ')
99987 Format (/1X,70('-')/1X,'Level : ',I10,'; output is ',I10,' by ',I10, &
        ' by ',I10/1X,70('-'))
99986 Format (/1X,A)
    End Program c09fcfe

```

10.2 Program Data

C09FCF Example Program Data

```

7, 6, 5           : m, n, fr
Bior1.1 period : wavnam, mode
3.0000 2.0000 2.0000 2.0000 1.0000 1.0000
2.0000 9.0000 1.0000 2.0000 1.0000 3.0000
2.0000 5.0000 1.0000 2.0000 1.0000 1.0000
1.0000 6.0000 2.0000 2.0000 7.0000 2.0000
5.0000 3.0000 2.0000 2.0000 4.0000 7.0000
2.0000 2.0000 1.0000 1.0000 2.0000 1.0000
6.0000 2.0000 1.0000 3.0000 6.0000 9.0000

2.0000 1.0000 5.0000 1.0000 2.0000 3.0000
2.0000 9.0000 5.0000 2.0000 1.0000 2.0000
2.0000 3.0000 2.0000 7.0000 1.0000 1.0000
2.0000 1.0000 1.0000 2.0000 3.0000 1.0000
2.0000 1.0000 2.0000 8.0000 3.0000 3.0000
1.0000 4.0000 5.0000 1.0000 2.0000 7.0000
8.0000 1.0000 3.0000 9.0000 1.0000 2.0000

3.0000 1.0000 4.0000 1.0000 1.0000 1.0000
1.0000 1.0000 2.0000 1.0000 2.0000 6.0000
4.0000 1.0000 7.0000 2.0000 5.0000 6.0000

```

```

3.0000  2.0000  1.0000  5.0000  9.0000  5.0000
1.0000  1.0000  2.0000  2.0000  2.0000  1.0000
2.0000  6.0000  3.0000  9.0000  5.0000  1.0000
1.0000  1.0000  8.0000  2.0000  1.0000  3.0000

5.0000  8.0000  1.0000  2.0000  2.0000  1.0000
1.0000  2.0000  2.0000  9.0000  2.0000  9.0000
2.0000  2.0000  2.0000  1.0000  1.0000  3.0000
1.0000  1.0000  1.0000  5.0000  1.0000  2.0000
3.0000  2.0000  8.0000  1.0000  9.0000  2.0000
2.0000  1.0000  9.0000  1.0000  2.0000  2.0000
3.0000  6.0000  5.0000  3.0000  2.0000  2.0000

5.0000  2.0000  1.0000  2.0000  1.0000  1.0000
3.0000  1.0000  9.0000  1.0000  2.0000  1.0000
2.0000  3.0000  1.0000  1.0000  7.0000  2.0000
7.0000  2.0000  2.0000  6.0000  1.0000  1.0000
5.0000  1.0000  7.0000  2.0000  1.0000  1.0000
2.0000  1.0000  3.0000  2.0000  2.0000  1.0000
5.0000  3.0000  9.0000  1.0000  4.0000  1.0000

```

10.3 Program Results

C09FCF Example Program Results

```

MLDWT :: Wavelet : Bior1.1
        End mode : period
        M       :    7
        N       :    6
        FR      :    5

```

```

Input Data      A :
Frame 1 :
3.0000  2.0000  2.0000  2.0000  1.0000  1.0000
2.0000  9.0000  1.0000  2.0000  1.0000  3.0000
2.0000  5.0000  1.0000  2.0000  1.0000  1.0000
1.0000  6.0000  2.0000  2.0000  7.0000  2.0000
5.0000  3.0000  2.0000  2.0000  4.0000  7.0000
2.0000  2.0000  1.0000  1.0000  2.0000  1.0000
6.0000  2.0000  1.0000  3.0000  6.0000  9.0000
Frame 2 :
2.0000  1.0000  5.0000  1.0000  2.0000  3.0000
2.0000  9.0000  5.0000  2.0000  1.0000  2.0000
2.0000  3.0000  2.0000  7.0000  1.0000  1.0000
2.0000  1.0000  1.0000  2.0000  3.0000  1.0000
2.0000  1.0000  2.0000  8.0000  3.0000  3.0000
1.0000  4.0000  5.0000  1.0000  2.0000  7.0000
8.0000  1.0000  3.0000  9.0000  1.0000  2.0000
Frame 3 :
3.0000  1.0000  4.0000  1.0000  1.0000  1.0000
1.0000  1.0000  2.0000  1.0000  2.0000  6.0000
4.0000  1.0000  7.0000  2.0000  5.0000  6.0000
3.0000  2.0000  1.0000  5.0000  9.0000  5.0000
1.0000  1.0000  2.0000  2.0000  2.0000  1.0000
2.0000  6.0000  3.0000  9.0000  5.0000  1.0000
1.0000  1.0000  8.0000  2.0000  1.0000  3.0000
Frame 4 :
5.0000  8.0000  1.0000  2.0000  2.0000  1.0000
1.0000  2.0000  2.0000  9.0000  2.0000  9.0000
2.0000  2.0000  2.0000  1.0000  1.0000  3.0000
1.0000  1.0000  1.0000  5.0000  1.0000  2.0000
3.0000  2.0000  8.0000  1.0000  9.0000  2.0000
2.0000  1.0000  9.0000  1.0000  2.0000  2.0000
3.0000  6.0000  5.0000  3.0000  2.0000  2.0000
Frame 5 :
5.0000  2.0000  1.0000  2.0000  1.0000  1.0000
3.0000  1.0000  9.0000  1.0000  2.0000  1.0000
2.0000  3.0000  1.0000  1.0000  7.0000  2.0000
7.0000  2.0000  2.0000  6.0000  1.0000  1.0000
5.0000  1.0000  7.0000  2.0000  1.0000  1.0000

```



```

2.0000  1.0000  3.0000  2.0000  2.0000  1.0000
5.0000  3.0000  9.0000  1.0000  4.0000  1.0000

```

```

Number of Levels :          2
Number of coefficients in 1st dimension for each level :
    2          4
Number of coefficients in 2nd dimension for each level :
    2          3
Number of coefficients in 3rd dimension for each level :
    2          3

```

```

-----
Level :          1; output is          4 by          3 by          3
-----

```

```

Detail coefficients (HLL)
Level 1, Coefficients 4 :
Frame 1 :
-4.9497  0.0000  0.0000
 0.7071  1.7678 -3.1820
 0.7071  2.1213  1.7678
 0.0000  0.0000  0.0000
Frame 2 :
 4.2426 -2.1213 -4.9497
 0.7071 -0.0000 -0.7071
-1.4142 -3.1820  1.4142
 0.0000  0.0000  0.0000
Frame 3 :
 2.1213 -4.9497 -0.7071
-2.8284 -4.2426  4.9497
 2.1213  2.8284 -0.7071
 0.0000  0.0000  0.0000

```

```

Success: the reconstruction matches the original.

```
